

```

100 Y2 = 2 + 1 / 16 / 12: REM The distance (in feet) the ball traveled between the
1st and 2nd light beam
110 Y3 = 5 + 3 / 16 / 12: REM The distance (in feet) the ball traveled between the
2nd and 3rd light beam

```

```

300 POKE - 16296,0: REM Turn off AND
310 POKE - 16294,0: REM Turn off AN1
320 POKE - 16292,0: REM Turn off AN2
330 POKE - 16290,0: REM Turn off AN3

```

```

400 PRINT CHR$(4) + "PR#3": REM Make text small (80 columns)

```

```

3320 PRINT "Press: 0)end 1)do the ball drop 2)clear"
3330 PRINT " 3)calibrate how many computer ticks in each sec. "
3340 PRINT " 4) enter computer ticks per sec. "
3345 PRINT " 5) enter (y2) and (y3) ": GET Z$: PRINT : PRINT
3350 IF Z$ = "2" THEN GOTO 300
3360 IF Z$ = "0" THEN CALL 976: END
3370 IF Z$ = "3" THEN GOTO 5000
3380 IF Z$ = "1" THEN GOTO 5000
3390 IF Z$ = "4" THEN GOTO 4000
3395 IF Z$ = "5" THEN GOTO 9000
3400 GOTO 400

```

```

4000 PRINT
4010 PRINT "Computer ticks per sec (ks) = ":KS
4020 INPUT "Enter new value for ks:":A$
4025 IF A$ < > "" THEN KS = VAL (A$)
4030 GOTO 400

```

```

5000 PRINT "DOING THE BALL DROP OR CALIBRATE COMPUTER TICKS PER SEC."
5002 IF Z$ = "3" THEN INPUT "How many sec. between BTN press? ":L
5003 IF Z$ = "3" THEN I% = - 32767
5004 IF Z$ = "1" THEN I% = 0
5005 K1% = I%
5010 IF PEEK ( - 16287) < 128 THEN 5010
5015 POKE - 16295,0: REM Turn on AND
5020 I% = I% + 1
5030 IF PEEK ( - 16286) < 128 THEN GOTO 5020
5040 K2% = I%
5045 POKE - 16293,0: REM Turn on AN1
5050 I% = I% + 1
5060 IF PEEK ( - 16285) < 128 THEN GOTO 5050
5070 K3% = I%
5075 POKE - 16291,0: REM Turn on AN2
5999 IF Z$ = "1" THEN 7000

```

```

6000 REM *** Calculating KS using K3% ***
6030 PRINT "K1=":K1%:" K2=":K2%:" K3=":K3%:" "
6050 KS = (K3% / L - K1% / L) / L
6060 PRINT "KS = ":KS
6070 PRINT "****Please press a key to go on****": GET A$
6072 POKE - 16296,0
6075 POKE - 16294,0
6077 POKE - 16292,0
6080 POKE - 16290,0
6999 GOTO 8115

```

```

7000 REM *** Ball Drop Calculations ***
7010 Y1 = 0: REM The distance (in feet) the ball traveled after breaking the first
beam
7040 T1 = 0.000: REM the time, in sec., when the ball breaks the 1st beam
7050 T2 = K2% / KS: REM the time the ball breaks the 2nd beam
7060 T3 = K3% / KS: REM the time the ball breaks the 3rd beam
7070 G = 2 * ((Y3 / T3) - (Y2 / T2)) / (T3 - T2): REM acceleration of gravity
7080 V0 = 0: REM The speed (ie.Velocity), in feet/second, of the ball when you let
go of the ball
7090 V1 = Y2 / T2 - G * T2 / 2: REM The Velocity of the ball when it breaks the
1st beam
7100 V2 = V1 + G * T2: REM Velocity of ball at 2nd beam
7110 V3 = V1 + G * T3: REM Velocity of ball at 3rd beam
7120 Y0 = 0 - (V1 * V1) / (2 * G): REM how high above 1st beam ball was held
7130 YI = Y0 * 12: REM distance above 1st beam (in inches)
7140 T0 = 0 - SQR (2 * (0 - Y0) / G): REM time when person lets go of the ball

8000 REM *** Print the Values ***
8010 PRINT "G=";G;" feet per (sec * sec); gravitational constant"
8030 PRINT "K1=";K1%;" K2=";K2%;" K3=";K3%;" tick counts"
8060 PRINT "KS=";KS;" computer ticks per sec."
8080 PRINT "T0=";T0;" the time (in sec) when the ball is let go"
8082 PRINT "T1=";T1;" the time in sec when the ball breaks the 1st beam"
8084 PRINT "T2=";T2;" the time when the ball breaks the 2nd beam"
8086 PRINT "T3=";T3;" the time in sec the ball breaks the 3rd beam"
8090 PRINT "V0=";V0;" feet / sec when ball is let go"
8091 PRINT "V1=";V1;" feet / sec at 1st light beam"
8092 PRINT "V2=";V2;" feet / sec at 2nd light beam"
8093 PRINT "V3=";V3;" feet / sec at 3rd light beam"
8094 PRINT "Y0=";Y0;" the dist. (in ft) the ball traveled before the 1st beam"
8095 PRINT "YI=";YI;" The dist. in inches"
8100 PRINT "Y1=";Y1;" Y2=";Y2;" Y3=";Y3;" Dist. (in ft.) to beams"
8110 POKE - 16296,0: POKE - 16294,0: POKE - 16292,0: POKE 16290,0
8111 PRINT : PRINT "***Hit any key to go on***"; GET A$
8115 CALL - 936: GOTO 3320

9000 PRINT
9005 PRINT "The distance (Y2) is ";Y2;" (In Feet)"
9010 INPUT "What do you want (Y2) to be?: ";A$
9020 PRINT "The distance (Y3) is ";Y3;" (In Feet)"
9030 INPUT "What do you want (Y3) to be?: ";B$
9040 IF A$ < > "" THEN Y2 = VAL (A$)
9050 IF B$ < > "" THEN Y3 = VAL (B$)
9060 GOTO 300

9999 CALL 976: END

```